
1. Overview

This document lists TTCN changes needed to correct issues in the ATS ‘iwd-TTCN3-B2015-03_D16wk37.

3. Correction to noIMSoGERAN procedures
Change 1 – Correction to function ‘f_GERAN_RejectOptionalPDPContext’
	Function name
	f_GERAN_RejectOptionalPDPContext ()

	Reason for change
	Not all UE initiate a PDPContextActivation for IMS on GERAN. The PDPContextReject should only be sent when UE has initiated a PDPContextActivation.

	Summary of change
	In function f_GERAN_RejectOptionalPDPContext the PDPContextReject is only sent when a PDPContextActivation is initiated by UE.

	TTCN module
	Common/GERAN/GERAN_CommonFunctions.ttcn

	MCC160 Comment
	Accepted in principle, but introduced a flag which is set if the PDP Context is received either in the default behaviour or the function


Before change

...

  /*

   * @desc      Function used in Reject PDP Context to postpone UE doing IMS

   * @param     p_CellId

   * @status    APPROVED (LTE_IRAT)

   */

  function f_GERAN_RejectOptionalPDPContext(GERAN_CellId_Type p_CellId) runs on GERAN_PTC

  {

    var TI v_TI;

    var GERAN_CellId_Type v_LLMEId := p_CellId;

    var G_L2_DATAMESSAGE_IND v_ChanReq;

    var LLCUnitDataIndType v_PDPCxt;

    var template (omit) G_LLC_DATAMESSAGE_IND v_ULDataInd;

    timer t_WaitForIMS := 10.0;

    if (f_GERAN_MsgInDefault_CheckPDPCxt()) {

      v_PDPCxt := f_GERAN_MsgInDefault_GetPDPCxt ();

      v_TI := v_PDPCxt.actPDPContextReq.ti;

    } else {

      t_WaitForIMS.start;

      alt {

        [] t_WaitForIMS.timeout{} // do nothing as IMS and PDP Context not required

        [] G_L2.receive(car_G_L2_ACCESS_IND(p_CellId, tsc_PhyCh0, tsc_RACH, ?, ?, cr_ChanReqOnePhase)) -> value v_ChanReq

          {

            v_ULDataInd := f_GERAN_ULTBFOnePhase(p_CellId, v_ChanReq.access, car_G_LLC_UnitData_IND(cr_G_ActPDP_ContextReqMO_Any));

            v_TI := valueof(v_ULDataInd.unit.msg.actPDPContextReq.ti);

          }

      }

    }

    // Send Reject

    v_TI.tiFlag := '1'B;

    f_GPRS_SendDownlinkMsg(p_CellId, 1, cas_G_LLC_UnitData_Req(f_GERAN_TLLI_Get(p_CellId),

                                                               tsc_LLCSapi_GMM,

                                                               tsc_LLC_PM,

                                                               px_GSM_CipheringOnOff,

                                                               cs_G_PDPContextReject(v_TI),

                                                               v_LLMEId));
  }
...
After change

...

  /*

   * @desc      Function used in Reject PDP Context to postpone UE doing IMS

   * @param     p_CellId

   * @status    APPROVED (LTE_IRAT)

   */

  function f_GERAN_RejectOptionalPDPContext(GERAN_CellId_Type p_CellId) runs on GERAN_PTC

  {

    var TI v_TI;

    var GERAN_CellId_Type v_LLMEId := p_CellId;

    var G_L2_DATAMESSAGE_IND v_ChanReq;

    var LLCUnitDataIndType v_PDPCxt;

    var template (omit) G_LLC_DATAMESSAGE_IND v_ULDataInd;

    timer t_WaitForIMS := 10.0;

    if (f_GERAN_MsgInDefault_CheckPDPCxt()) {

      v_PDPCxt := f_GERAN_MsgInDefault_GetPDPCxt ();

      v_TI := v_PDPCxt.actPDPContextReq.ti;

    } else {

      t_WaitForIMS.start;

      alt {

        [] t_WaitForIMS.timeout{} // do nothing as IMS and PDP Context not required

        [] G_L2.receive(car_G_L2_ACCESS_IND(p_CellId, tsc_PhyCh0, tsc_RACH, ?, ?, cr_ChanReqOnePhase)) -> value v_ChanReq

          {

            v_ULDataInd := f_GERAN_ULTBFOnePhase(p_CellId, v_ChanReq.access, car_G_LLC_UnitData_IND(cr_G_ActPDP_ContextReqMO_Any));

            v_TI := valueof(v_ULDataInd.unit.msg.actPDPContextReq.ti);

            // Send Reject

            v_TI.tiFlag := '1'B;

            f_GPRS_SendDownlinkMsg(p_CellId, 1, cas_G_LLC_UnitData_Req(f_GERAN_TLLI_Get(p_CellId),

                                                               tsc_LLCSapi_GMM,

                                                               tsc_LLC_PM,

                                                               px_GSM_CipheringOnOff,

                                                               cs_G_PDPContextReject(v_TI),

                                                               v_LLMEId));

          }
      }

    }
  }
...
MCC160 Implementation:
  function f_GERAN_RejectOptionalPDPContext(GERAN_CellId_Type p_CellId) runs on GERAN_PTC

  {

    var TI v_TI;

    var GERAN_CellId_Type v_LLMEId := p_CellId;

    var G_L2_DATAMESSAGE_IND v_ChanReq;

    var LLCUnitDataIndType v_PDPCxt;

    var template (omit) G_LLC_DATAMESSAGE_IND v_ULDataInd;

    var boolean v_PDPCxtRxd := false; // @sic R5s160890 sic@

    timer t_WaitForIMS := 10.0;

    if (f_GERAN_MsgInDefault_CheckPDPCxt()) {

      v_PDPCxt := f_GERAN_MsgInDefault_GetPDPCxt ();

      v_TI := v_PDPCxt.actPDPContextReq.ti;

      v_PDPCxtRxd := true; // @sic R5s160890 sic@

    } else {

      t_WaitForIMS.start;

      alt {

        [] t_WaitForIMS.timeout{} // do nothing as IMS and PDP Context not required

        [] G_L2.receive(car_G_L2_ACCESS_IND(p_CellId, tsc_PhyCh0, tsc_RACH, ?, ?, cr_ChanReqOnePhase)) -> value v_ChanReq

          {

            v_ULDataInd := f_GERAN_ULTBFOnePhase(p_CellId, v_ChanReq.access, car_G_LLC_UnitData_IND(cr_G_ActPDP_ContextReqMO_Any));

            v_TI := valueof(v_ULDataInd.unit.msg.actPDPContextReq.ti);

            v_PDPCxtRxd := true; // @sic R5s160890 sic@

          }

      }

    }

    if (v_PDPCxtRxd) { // @sic R5s160890 sic@

      // Send Reject

      v_TI.tiFlag := '1'B;

      f_GPRS_SendDownlinkMsg(p_CellId, 1, cas_G_LLC_UnitData_Req(f_GERAN_TLLI_Get(p_CellId),

                                                                 tsc_LLCSapi_GMM,

                                                                 tsc_LLC_PM,

                                                                 px_GSM_CipheringOnOff,

                                                                 cs_G_PDPContextReject(v_TI),

                                                                 v_LLMEId));

    }
  }

